
The helium ground state  1S2 is worked within the Hartree Fock method ,by casting it into the Roothaan equations  matrix frame.

Reference:

1.  Modern Quantum Chemistry : Introduction to Advanced Electronic Structure Theory (Paperback) 
by Attila Szabo, Neil S. Ostlund , Chapter 3.

2.C. C. J. Roothaan, New developments in molecular orbital theory, Re. Mod. Phys. 23 : 69 (1951).
3. ab initio Methods 
Hartree-Fock-Roothan Equations http://core.ecu.edu/phys/flurchickk/AtomicMolecularSystems/methodsDocumentation/qmnotes/node3.html
4. The Hartree-Fock Method for Atoms by Charlotte Froese Fischer 
Consider the helium atom ground state as  represented by the product of two 1s wave functions 
Ψ (1,2) = ψ1s(1)ψ1s(2)        .                                (1)

ψ1s(1) is expanded in terms of a basis set   { φ1 φ2….. φN }. In the present example we use a normalized  ( but not orthogonal) two basis set of the Slater type orbitals (STO),
ψ1s(1) = c1 φ1 +  c2 φ2                                          (2)

where  φi = (  αi3 / π)1/2 exp(- αi r)    . 

Roothaan equations have the compact matrix form
    F C  = S C ε                                                (3)
Where F  is the matrix of the one electron Hamiltonian

Fij = < φi  / (-1/2) ∆ - Z / r   + Φ ee (r)  / φj >   ,   ( 4  )

Φ ee (r)  = ∫  ψ 2 (r’)  dτ’ / abs(r-r’)             ,          ( 5 )
and Sij = < φi    / φ j  >   (the overlap matrix)   .             (6)
The coefficients matrix has n columns corresponding to n eigenvectors. To the first eigenvalue  ε1 , corresponds the first column with elements   c11 , c21, c31,….cn1. To the second  ε2 , corresponds the second column  c12,c22,c32….cn2.
C = 

                                                ( 7 )
It is through ψ 2 (r’)  in (5) that the self consistency has to be achieved. The first choice for ψ in the intgrand is  
ψ (x) = ( α13 / π)1/2 exp(- α1  r)      ,                           (8)

where a good value of    α1  would be around Z-5./16.
ε is a the diagonal energy  matrix with eigenvalues  ε1 ,

ε 2 , ε3 …..
Using a two basis set , equation (3) leads to the characteristic equation is   

( 1- S2 12  ) E2   +( 2F12 S12  - (F11 +F22 )  )E  
+ F11 F 22  - (F12) 2       =0                                 (9)
or      A E2 + B E + C    =0.   
The roots are   
E1 =  {- B – ( B2 -4AC )1/2 } /(2A)                      (10  )   

E2 =  {- B + ( B2 -4AC )1/2 } /(2A)                        (11)

The unormalized eigenvectors are taking c(1,1)=1.

and also c(1,2) =1   ,

c(1,1)  =1

c(2,1)  =  - (F11 - E1) /( F12 - E1 S 12 )                (12 )

and 

c(1,2)=1

c(2,2)  =  - (F11 – E2) /( F12 – E2 S 12 )   .             (13 )

The total energy is given by
E = 2 ε1s  -   

( c1 φ1 +  c2 φ2)2  Φ ee(r) dτ  .      (14)

The Hartree Fock total energy value is, to five digits, E = – 2.8617 au and the orbital energy ε1s = -0.91796
Here we obtain E = - 2.8600 au.

The last five iterations out of ten, are shown next.

 ve11,ve12,ve22=  1.12698686  0.869139194  0.811295748

 h11,h12,h22,s12=  -0.8730E+00  -0.8784E+00  -0.7773E+00   0.8738E+00

 e,coef11,coef21= -0.916404903  1.  0.558482766

 alfa,alfa2,etotal=    0.20000E+01    0.10929E+01   -0.28603E+01

 ve11,ve12,ve22=  1.12744868  0.869440794  0.811503291

 h11,h12,h22,s12=  -0.8726E+00  -0.8781E+00  -0.7771E+00   0.8738E+00

 e,coef11,coef21= -0.916027546  1.  0.559367299

 alfa,alfa2,etotal=    0.20000E+01    0.10929E+01   -0.28598E+01

 ve11,ve12,ve22=  1.12731969  0.869347394  0.811414957

 h11,h12,h22,s12=  -0.8727E+00  -0.8782E+00  -0.7772E+00   0.8738E+00

 e,coef11,coef21= -0.91614157  1.  0.559219658

 alfa,alfa2,etotal=    0.20000E+01    0.10929E+01   -0.28600E+01

 ve11,ve12,ve22=  1.12733328  0.869355798  0.811422169

 h11,h12,h22,s12=  -0.8727E+00  -0.8782E+00  -0.7772E+00   0.8738E+00

 e,coef11,coef21= -0.916130543  1.  0.559243798

 alfa,alfa2,etotal=    0.20000E+01    0.10929E+01   -0.28600E+01
 ve11,ve12,ve22=  1.12732983  0.869353831  0.811420858

 h11,h12,h22,s12=  -0.8727E+00  -0.8782E+00  -0.7772E+00   0.8738E+00

 e,coef11,coef21= -0.916133225  1.  0.559236526

 alfa,alfa2,etotal=    0.20000E+01    0.10929E+01   -0.28600E+01
FORTRAN CODE

c Roothan Equations  Ref Szabo and Ostlund Chapter 3.

c psi=c1*sqrt(alfa1**3/pi)* exp(-alfa1*r) +c2*sqrt(alfa2**3/pi)*

c   exp(-alfa2*r)

c Roothaan Equations--example of  Helium ground state  1s^2

      dimension h(2,2), s(2,2) , coef(2,2), vee(2,2), vecoul(0:10000)

      pi=2.*asin(1.)

      Z=2.

      alfa1=z

      nalfa=1

      niter=10

      alfafin=2.5

      dalfa=(alfafin-alfa1)/float(nalfa)

      do 10 ialfa=1,nalfa

      alfa2=alfa1/(1.83)

      s(1,1)=1.

      s(1,2)= 8.*(alfa1*alfa2)**(1.5)/(alfa1+alfa2)**3

      s(2,1)=s(1,2)

      s(2,2)=1.

      xlim=9.

      nstep=7000

      dx=xlim/float(nstep)

      do 70 iter=1,niter

      if(iter.eq.1)then

      an1s=1.

      an2s=1.

      endif

      call ve(iter,alfa1,alfa2,an1s,pi,xlim,nstep,coef,vecoul)

       call veeij(iter,xlim,nstep,alfa1,alfa1,alfa1,pi,coef,

     $ an1s,an2s,vecoul,veij)

      vee(1,1)=veij

      call veeij (iter,xlim,nstep,alfa1,alfa1,alfa2,pi,coef,

     $ an1s,an2s,vecoul, veij)

      vee(1,2)=veij

      vee(2,1)=vee(1,2)

      call veeij(iter,xlim,nstep,alfa1,alfa2,alfa2,pi,

     $coef,an1s,an2s,vecoul,veij)

      vee(2,2)=veij

      print*,'ve11,ve12,ve22=',vee(1,1),vee(1,2),vee(2,2)

      h(1,1)= alfa1**2/2. - Z*alfa1 + vee(1,1)

      t12a=-4.*alfa2**2*(alfa1*alfa2)**(1.5)/(alfa1+alfa2)**3

      t12b= 4.*alfa2*(alfa1*alfa2)**(1.5)/(alfa1+alfa2)**2

      t12=t12a+t12b

      v12=-4.*Z*(alfa1*alfa2)**(1.5)/(alfa1+alfa2)**2

      h(1,2)= t12 + v12 +vee(1,2)

      h(2,1)=h(1,2)

      h(2,2)= alfa2**2/2. - Z*alfa2 + vee(2,2)

      A=1.-S(1,2)**2

      B=2.*h(1,2)*S(1,2)-(h(1,1)+h(2,2))

      C=h(1,1)*h(2,2)-h(1,2)**2

      E2=(-b+sqrt(b**2-4.*a*c) )/(2.*a)

      e1= (-b-sqrt(b**2-4.*a*c) )/(2.*a)

      if(iter.eq.1)ajcoul=vee(1,1)

      call coefi(h,s,E1,1,coef)

      call coefi(h,s,E2,2,coef)

      call plotwv(xlim,nstep,alfa1,alfa2,pi,coef,an1s,an2s)

      if(iter.gt.1)call vrep(nstep,dx,pi,coef,an1s,alfa1,alfa2,

     $ vecoul,ajcoul)

      etotal=2.*e1-ajcoul

      print 100,alfa1,alfa2,etotal

      print*,'   '

70    continue

      alfa1=alfa1 + dalfa

10    continue

100   format(1x,'alfa,alfa2,etotal=',3(3x,e12.5))

      if(nalfa.eq.1)then

c      call coefi(h,s,E1,1,coef)

      print*,'  '

c      call coefi(h,s,E2,2,coef)

      endif

      stop

      end

      Subroutine coefi(h,s,E,neigen,coef)

      dimension h(2,2), s(2,2),coef(2,2)

      print 100,h(1,1),h(1,2),h(2,2),s(1,2)

      coef(1,neigen)=1.

      anum=h(1,1)-e

      if(abs(anum).le.1.e-5)then

      coef(2,neigen)=0.

      goto 50

      endif

      coef(2,neigen)=-(h(1,1)-e)*coef(1,neigen)/(h(1,2)-e*s(1,2) )

50    print*,'e,coef11,coef21=',e, coef(1,neigen),coef(2,neigen)

100   format(1x,'h11,h12,h22,s12=',4(2x,e11.4))

      return

      end

      Subroutine plotwv(xlim,nstep,alfa1,alfa2,pi,coef,an1s,an2s)

      dimension coef(2,2)

      phi1(x)=sqrt(alfa1**3/pi)*exp(-alfa1*x)

      phi2(x)=sqrt(alfa2**3/pi)*exp(-alfa2*x)

      psi1sn(x)=an1s*( coef(1,1)*phi1(x)+coef(2,1)*phi2(x) )

      psi2sn(x)=an2s*( coef(1,2)*phi1(x)+coef(2,2)*phi2(x) )

      dx=xlim/float(nstep)

      sum1s=0.

      sum2s=0.

      do 10 i=1,nstep

      x=dx*float(i)

      sum1s=sum1s+ ((coef(1,1)*phi1(x)+coef(2,1)*phi2(x))*x)**2 +

     $ ( ( coef(1,1)*phi1(x-dx)+coef(2,1)*phi2(x-dx) )*(x-dx))**2

        sum2s=sum2s+ ((coef(1,2)*phi1(x)+coef(2,2)*phi2(x) )*x)**2 +

     $ ( (coef(1,2)*phi1(x-dx)+coef(2,2)*phi2(x-dx) )*(x-dx))**2

10    continue

      sum1s=4.*pi*sum1s*dx/2.

      sum2s=4.*pi*sum2s*dx/2.

      an1s=1./sqrt(sum1s)

      an2s=1./sqrt(sum2s)

c      print*,'   '

c      do  20 i=0,nstep,20

c      x=dx*float(i)

c      print 100,x, psi1sn(x), psi2sn(x)

c20    continue

100   format(1x,'x,psi1s,psi2s=',3(3x,e11.4))

      return

      end

      subroutine veeij(it,xlim,nstep,alfa,alfaone,alfatwo,pi,coef,

     $ an1s,an2s,vecoul,veij)

      dimension coef(2,2) , vecoul(0:10000)

       vee(x) = alfa*( exp(-2.*alfa*x)*(-1.-1./(alfa*x) ) +

     $ 1./(alfa*x))

       phi(al,x)= sqrt(al**3/pi)*exp(-al*x)

       f(x)=phi(alfaone,x)*phi(alfatwo,x)*vee(x)*x**2

       phicf(x) =an1s*(coef(1,1)*phi(alfaone,x) +coef(2,1)*

     $  phi(alfatwo,x) )

       sum=0.

       dx=xlim/float(nstep)

       if(it.eq.1)then

       do 10 i=1,nstep

       x=dx*float(i)

       sum=sum + f(x) + f(x-dx+1.e-5)

10    continue

      sum=sum*4.*pi*dx/2.

      veij=sum

c      print*,'veij=',veij

      endif

      if(it.gt.1)then

      sum=0.

      do 20 i=1,nstep

      x=dx*float(i)

      sum=sum+ phi(alfaone,x)*phi(alfatwo,x)*vecoul(i)*x**2+

     $ phi(alfaone,x-dx)*phi(alfatwo,x-dx)*vecoul(i-1)*(x-dx)**2

20    continue

      sum=sum*4.*pi*dx/2.

      veij=sum

c      print*,'it,vecoul(0),veij=',it,vecoul(0),veij

      endif

      return

      end

      subroutine ve(iter,alfa1, alfa2,an1s,pi,xlim,nstep,coef,vecoul)

c creates vecoul  for the iterations

      dimension vecoul(0:10000),coef(2,2)

      psi1(x)= sqrt(alfa1**3/pi)*exp(-alfa1*x)

      psi2(x)= sqrt(alfa2**3/pi)*exp(-alfa2*x)

      psi(x)=an1s*(coef(1,1)*psi1(x)+coef(2,1)*psi2(x) )

      rho(x)= psi(x)**2

c     calculates vecoul(0)

c      print*,'subroutine ve,alfa1,an1s,coef(1,1),cf12=',alfa1,an1s,

c     $coef(1,1),coef(2,1)

      dx=xlim/float(nstep)

      sum=0.

      do 10 i=1,nstep

      x= dx*float(i)

      if(iter.eq.1)sum=sum + psi1(x-dx)**2*(x-dx) + psi1(x)**2*x

      if(iter.ge.2)sum=sum + rho(x-dx)*(x-dx) + rho(x)*x

10    continue

      sum=sum*4.*pi*dx/2.

      vecoul(0)=sum

      vecoul(1)=vecoul(0)

      do 20 i=2,nstep

      x=dx*float(i)

      vecoul(i)=2.*vecoul(i-1)-vecoul(i-2) - (2/(x-dx))*dx*

     $ (vecoul(i-1)-vecoul(i-2))

     $ - 4.*pi*dx**2*rho(x-dx)

20    continue

      return

      end

      subroutine vrep(nstep,dx,pi,coef,an1s,alfa1,alfa2,vecoul,ajcoul)

      dimension coef(2,2), vecoul(0:10000)

      base1(x)=sqrt(alfa1**3/pi)*exp(-alfa1*x)

      base2(x)=sqrt(alfa2**3/pi)*exp(-alfa2*x)

      phi(x)=an1s*(coef(1,1)*base1(x)+coef(2,1)*base2(x) )

      f(x,i)=phi(x)**2*vecoul(i)*x**2

      sum=0.

      do 10 i=1,nstep

      x=dx*float(i)

      sum=sum +f(x-dx,i-1) +f(x,i)

10    continue

      sum=sum*4.*pi*dx/2.

      ajcoul=sum

      return

      end

